

Esercitazione 4

Istruzioni trattate: l'operatore \$, xgrid, legend, plot2d.

In questa esercitazione si considera il problema di approssimare lo *zero*, α , della funzione f definita per $x > 0$ da:

$$f(x) = x + \log x$$

Nella prima parte si confrontano le approssimazioni ottenute realizzando i metodi ad un punto definiti dalle funzioni:

$$h_2(x) = e^{-x} \quad \text{e} \quad h_3(x) = \frac{x + e^{-x}}{2}$$

Nella seconda parte si confrontano queste approssimazioni con quella ottenuta realizzando il *metodo di bisezione* applicato alla funzione f e quella ottenuta realizzando il metodo ad un punto definito dalla funzione:

$$h_4(x) = x - (f(x))^2$$

Prima parte

- L'operatore \$.

Sia v un vettore di n elementi. Le componenti di v si indicano con $v(1), \dots, v(n)$. L'*ultima* componente si può indicare anche con $v(\$)$. Se g è una *function*, l'assegnamento:

$$v(\$ + 1) = g(v(\$))$$

ha l'effetto di *creare una nuova componente* di v (la $n + 1$ - esima) ed assegnarle il valore $g(v(n))$.

La definizione che segue è una realizzazione del *metodo ad un punto* definito dalla funzione h .

```
function x = MetodoUnPunto(h,x0,N)
//
// Metodo ad un punto definito dalla funzione h, continua.
// La costruzione della successione si arresta dopo N iterazioni.
//
// x: colonna di numeri reali. Le componenti di questo vettore sono i primi
//   N + 1 elementi della successione generata dal metodo definito
//   da h a partire da x0.
//
x = x0;
k = 0;
while k < N,
    x($+1) = h(x($));
    k = k + 1;
end;
endfunction
```

Si osservi che la condizione che si è scelto di utilizzare per realizzare il criterio di arresto (arresta la costruzione della successione dopo aver eseguito N iterazioni) è certamente *calcolabile* ed *efficace* ma, quando verificata, *non* fornisce alcuna informazione sull'accuratezza dell'approssimazione trovata.

Come primo metodo per approssimare α , si considera quello definito dalla funzione h_2 . Lo studio *analitico* del metodo consente di scoprire che h_2 ha *un solo punto unito* (lo zero α) separato dall'intervallo $[\frac{1}{2}, 1]$, che il metodo è *utilizzabile* ed ha *ordine di convergenza uno* ad α e che la successione generata a partire da $x_0 = \frac{1}{2}$ converge ad α . Sussistono anche le seguenti *proprietà qualitative*: ciascun elemento della successione dista da α *meno* del precedente e si trova "dalla parte opposta" di α rispetto al precedente. Vediamo che informazioni si possono ottenere, invece, per via *grafica*.

Si consideri la seguente realizzazione della funzione h_2 :

```
function y = h2(x)
//
// x,y: matrici ad elementi reali di uguale dimensione;
//
y = exp(-x);
endfunction
```

Riportiamo su uno stesso piano cartesiano i grafici, sull'intervallo $[0, 1]$, delle funzioni $y = x$ e $y = h_2(x)$:¹

```
-->x = linspace(0,1,301)';
-->plot2d(x, [x,h2(x)]);
-->xgrid(3); xlabel('x'); legend('y = x', 'y = h2(x)', 'in_lower_right');
```

Si ottiene il disegno riportato in Figura 1. Il valore 3 dell'argomento del comando `xgrid` cambia il colore con cui viene disegnata la griglia da nero (colore predefinito) a verde. L'opzione `'in_lower_right'` presente nel comando `legend` fa posizionare la legenda nell'angolo *in basso a destra* della figura.




Figura 1: Grafico per lo studio del metodo definito da h_2 .

Dal disegno si deduce che, nell'intervallo $[0, 1]$: (a) i grafici di $h_2(x)$ ed x si intersecano in un solo punto, di ascissa maggiore di 0.5; (b) in tutti i punti del grafico di h_2 con ascissa maggiore o uguale a 0.5, la retta tangente ha pendenza negativa e maggiore di meno uno e (c) il punto unito è più vicino a 0.5 che a 1. Ne segue che: (a) la funzione h_2 ha un solo punto unito in $[0, 1]$, separato dall'intervallo $[\frac{1}{2}, 1]$; (b) l'intervallo $[\frac{1}{2}, 1]$ verifica le ipotesi del Teorema di convergenza dunque il metodo definito da h_2 è utilizzabile per l'approssimazione e (c) la successione generata dal metodo a partire da $x_0 = \frac{1}{2}$ risulta convergente al punto unito.

Si osservi che dal disegno si possono dedurre informazioni relative al *solo* intervallo $[0, 1]$. In particolare, il disegno *non fornisce informazioni sull'esistenza di altri punti uniti al di fuori di tale intervallo*. La scelta dell'intervallo da considerare per lo studio è *responsabilità dell'utilizzatore*.

Con le informazioni ottenute, analiticamente o graficamente, possiamo utilizzare la procedura `MetodoUnPunto` per ottenere un'approssimazione del punto unito α :

```
-->x0 = 1/2; N = 20;
-->z2 = MetodoUnPunto(h2,x0,N);
-->plot2d(z2($),h2(z2($)),style = -5);
```

¹Più correttamente: *un'approssimazione* di tali grafici.

```
-->legend('y = x','y = h2(x)','(z2 , h2(z2))','in_lower_right');
```

Si ottiene nella finestra grafica il disegno riportato in Figura 2. Il valore negativo per l'opzione `style` del comando `plot2d` fa disegnare *un simbolo* con centro nelle coordinate del punto richiesto (in questo caso il simbolo \diamond con centro in $(z2(\$), h2(z2(\$)))$). Si osserva che l'ultimo elemento della




Figura 2: L'approssimazione ottenuta con il metodo definito da h_2 .

successione generata approssima il punto unito con sufficiente “accuratezza grafica.”

Si considera adesso il metodo definito dalla funzione h_3 . Lo studio *analitico* del metodo consente di scoprire che h_3 ha *un solo punto unito* (lo zero α) separato dall'intervallo $[\frac{1}{2}, 1]$, che il metodo è *utilizzabile* ed ha *ordine di convergenza uno ad α* e che la successione generata a partire da $x_0 = \frac{1}{2}$ converge ad α . Sussiste anche la seguente *proprietà qualitativa*: la successione è *monotona crescente*. Anche in questo caso vediamo che informazioni si possono ottenere per via *grafica*.

Si consideri la seguente realizzazione della funzione h_3 :

```
function y = h3(x)
//
// x,y: matrici ad elementi reali di uguale dimensione;
//
y = (x + exp(-x)) / 2;
endfunction
```

Procedendo come per il metodo precedente si ottiene inizialmente il disegno riportato nella Figura 3.




Figura 3: Grafico per lo studio del metodo definito da h_3 .

Dal disegno si deduce che, nell'intervallo $[0, 1]$: (a) i grafici di $h_3(x)$ ed x si intersecano in un solo punto, di ascissa maggiore di 0.5; (b) in tutti i punti del grafico di h_3 la retta tangente ha pendenza positiva e minore di uno e (c) il punto unito è più vicino a 0.5 che a 1. Ne segue che: (a) la funzione h_3 ha un solo punto unito in $[0, 1]$, separato dall'intervallo $[\frac{1}{2}, 1]$; (b) l'intervallo $[\frac{1}{2}, 1]$ verifica le ipotesi del Teorema di convergenza dunque il metodo definito da h_3 è utilizzabile per l'approssimazione e (c) la successione generata dal metodo a partire da $x_0 = \frac{1}{2}$ risulta convergente al punto unito.

Come già osservato, dal disegno si possono dedurre informazioni relative al *solo* intervallo $[0, 1]$.

Con le informazioni ottenute, analiticamente o graficamente, possiamo utilizzare la procedura **MetodoUnPunto** per ottenere una nuova approssimazione del punto unito α e poi il disegno riportato in Figura 4. Anche in questo caso si osserva che l'ultimo elemento della successione generata




Figura 4: L'approssimazione ottenuta con il metodo definito da h_3 .

approssima il punto unito con sufficiente "accuratezza grafica."

Per confrontare graficamente l'andamento delle porzioni di successione determinate dai due metodi, se ne tracciano sullo stesso piano cartesiano i grafici.² Si ottiene il disegno riportato in Figura 5.




Figura 5: I grafici delle porzioni delle successioni ottenute.

²Il grafico della porzione di successione è costituito dai soli *vertici* delle spezzate. I lati sono mantenuti per chiarezza grafica.

Il disegno conferma quanto già noto dallo studio analitico riguardo al comportamento qualitativo delle successioni.

Un grafico più significativo si ottiene confrontando gli elementi delle successioni con α . Per poter effettuare il confronto è però necessario utilizzare un'approssimazione di α . Come già osservato, una scelta ragionevole è di approssimare α con $z3(\$)$. Si consideri allora il disegno riportato a sinistra nella Figura 6 generato dai comandi seguenti:³

```
-->a = z3($);
-->clf(); plot2d([0:N]',[abs(z2 - a),abs(z3 - a)], style = [2,5]);
-->xgrid(); xlabel('k'); legend('| z2(k) - a |','| z3(k) - a |');
```




Figura 6: A sinistra: distanza da α delle successioni. A destra: distanza in scala *semilogaritmica*.

Il disegno evidenzia, almeno per k da 1 a 9, che a parità di indice k la distanza da α dell'elemento della successione generata dal metodo definito da h_3 è *minore* di quella della successione generata dal metodo definito da h_2 . Il disegno a destra nella stessa figura riporta gli stessi grafici ma in scala *semilogaritmica*⁴ e mostra che anche per k da 10 a 19 la distanza da α dell'elemento della successione generata dal metodo definito da h_3 è *minore* di quella della successione generata dal metodo definito da h_2 . I disegni *suggeriscono* che, come in effetti accade, la successione generata dal metodo definito da h_3 converga ad α *più rapidamente* di quella generata dal metodo definito da h_2 .

Nel disegno a destra si osserva che le curve blu e rossa sembrano *porzioni di retta*. Questo si spiega osservando che (a) se x_k è una successione convergente ad α e generata dal metodo definito da una funzione h che verifica le ipotesi del Teorema di convergenza allora *asintoticamente*:

$$|x_k - \alpha| \approx |h'(\alpha)|^k |x_0 - \alpha|$$

e quindi:

$$\log_{10} |x_k - \alpha| \approx k \log_{10} |h'(\alpha)| + \log_{10} |x_0 - \alpha|$$

e (b) il grafico in scala semilogaritmica di $|h'(\alpha)|^k |x_0 - \alpha|$ è una retta, di pendenza tanto più *negativa* quanto più *piccolo* è $|h'(\alpha)|$.

Seconda parte

Modificando opportunamente la procedura **Bisezione** descritta nell'Esercitazione 3, in particolare arrestando la costruzione della successione dopo N iterazioni indipendentemente dalla misura dell'intervallo, ed utilizzando la realizzazione ingenua **f** della funzione f , i comandi:

³Nel secondo comando compare l'operatore : (colon). Se n ed m sono due numeri interi con $n < m$ allora $[n:m]$ è la *riga* di componenti $n, n + 1, \dots, m - 1, m$.

⁴Il comando utilizzato per generare il grafico è: `plot2d('n1',[0:N]',[abs(z2 - a),abs(z3 - a)], style = [2,5]);`

```

-->zB = Bisezione(f,1/2,1,N);
-->clf();
-->plot2d('nl',[0:N],[abs(z2 - a),abs(z3 - a),abs(zB - a)],style = [2,5,1]);
-->xgrid(); xlabel('k'); legend('| z2(k) - a |','| z3(k) - a |','| zB(k) - a |');

```

creano la variabile `zB`, le assegnano come valore la colonna dei primi 21 elementi della successione generata dalla procedura `Bisezione` applicata ad `f` a partire dall'intervallo $[\frac{1}{2}, 1]$ e generano il grafico riportato in Figura 7:




Figura 7: Confronto tra le successioni generate dai metodi definiti da h_2, h_3 e dal metodo di bisezione.

Per la successione generata dal metodo di bisezione, il grafico *suggerisce* che la distanza del k -esimo elemento da α *non* decresce in modo monotono al crescere di k e che la convergenza ad α è *più lenta* rispetto a quella della successione generata dal metodo definito da h_3 ma ha rapidità *simile* a quella della successione generata dal metodo definito da h_2 . Infatti, il metodo di bisezione si comporta “come” un metodo ad un punto definito da una funzione h con $|h'(\alpha)| = \frac{1}{2}$ e $|h'_2(\alpha)| = e^{-\alpha} \approx 0.57$.

Si consideri adesso la funzione h_4 e ed il metodo ad un punto da essa definito. L'equazione $x = h_4(x)$ è equivalente all'equazione $f(x) = 0$ ed i punti uniti di h_4 sono gli zeri di f . Dunque è *ragionevole* considerare il metodo per approssimare α . Per la ricerca di valori di x_0 che garantiscano la convergenza della successione generata, però, il Teorema di convergenza *non è utilizzabile*. Infatti si ha $h'_4(\alpha) = 1 - 2f(\alpha)f'(\alpha) = 1$. Ricorriamo allora allo *studio grafico* del metodo. Con l'aiuto del grafico riportato in Figura 8 si intuisce che la successione generata a partire da $x_0 = 0.6$ risulta monotona decrescente e inferiormente limitata, dunque convergente.




Figura 8: Grafico per lo studio del metodo definito ha h_4 .

Definita:

```

function y = h4(x)
//
// x,y: matrici ad elementi reali di uguale dimensione;
//
y = x - (x + log(x)) .^2;
endfunction

i comandi:

-->x0 = 0.6; N = 20;

-->z4 = MetodoUnPunto(h4,x0,N);

-->clf();

-->plot2d('nl',[0:N]',[abs(z2 - a),abs(z3 - a),abs(z4 - a)],style = [2,5,1]);

-->xgrid(); xlabel('k'); legend('| z2(k) - a |','| z3(k) - a |','| z4(k) - a |');

```

generano il grafico riportato in Figura 9:




Figura 9: Confronto tra le successioni generate dai metodi definiti da h_2 , h_3 e h_4 .

Per la successione generata dal metodo definito da h_4 , il grafico *suggerisce* che la distanza del k -esimo elemento da α *decreisce* in modo monotono al crescere di k e che la convergenza ad α è molto *più lenta* sia rispetto a quella della successione generata dal metodo definito da h_3 che rispetto a quella della successione generata dal metodo definito da h_2 . In effetti si dimostra analiticamente che se x_k è una successione convergente ad α e generata da un metodo definito da h con $|h'(\alpha)| = 1$ allora la successione $|x_k - \alpha|$ tende a zero *più lentamente* della successione θ^k per ogni $\theta \in (0, 1)$.

Esercizi

1. Si consideri la definizione seguente:

```
function y = h4v(x)
    if x > 0 then
        y = x - ( x + log(x) ) .^2;
    else
        disp(x); error('*** argomento non positivo in h4v! ***')
    end
endfunction
```

e si eseguano i comandi:

```
-->disp(h4(-1))
```

```
-->disp(h4v(-1))
```

In *Scilab* la *funzione predefinita* `log` corrisponde alla funzione *logaritmo complesso*, definita per ogni numero complesso $z \neq 0$ da:

$$\log(z) = \log(|z|) + i \arg(z)$$

dove $\arg(z)$ è l'*argomento principale* di z (l'unico argomento di z in $(-\pi, \pi]$).⁵ La verifica sull'argomento nella *function* `h4v` ha lo scopo di evitare fraintendimenti: se l'argomento non è positivo, la funzione *deve* segnalarlo all'utilizzatore.

Si discuta infine l'esecuzione dei comandi:

```
-->x0 = 0.5; N = 4;
```

```
-->s4 = MetodoUnPunto(h4v,x0,N);
```

```
-->s4 = MetodoUnPunto(h4,x0,N);
```

```
-->s4($)
```

con l'aiuto del grafico riportato in Figura 10.




Figura 10: Grafico per l'Esercizio 1.

⁵Per approfondire, vedere, ad esempio: https://en.wikipedia.org/wiki/Logarithm#Complex_logarithm.